
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc122001091]Corrections required
1.1 [bookmark: _Toc122001092]Change 1
	Function name
	fl_TC_11_1_2_NR5GC_EUTRA_TestBody

	Reason for change
	With the implementation of the R5-216188, specifically parallel behaviour in table 11.1.2.3.2-2, the implementation in TTCN is hardconding the expected APN to be of pc_APN_ID_Internet. However depending on the UE setting of the default APN, it may also be of pc_APN_ID_IMS.

	Summary of change
	TTCN now accepts both APNs (pc_APN_ID_IMS and pc_APN_ID_Internet) in its implementation of the parallel behaviour of the table 11.1.2.3.2-2

	TTCN module
	NR5GC_IRAT\11_1\EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



    
Before Change:
	   function fl_TC_11_1_2_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC // @sic R5-204399 sic@
  {
    var integer v_EPSBearerId;

    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 5AA" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE);
    
    f_TAUReject_AttachWithoutN26 (eutra_Cell1);
    
    // @siclog "Parallel Behaviour" siclog@
    f_EUTRA38_MultiPDN_EstablishAdditionalPDNs (eutra_Cell1,
                                                (pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection),
                                                false,
                                                tsc_SRB2,
                                                omit,
                                                '010'B,
                                                cr_AccessPointName(f_DomainName_Encode(pc_APN_ID_Internet))); // @sic R5-216188 sic@


    // @siclog "Step 24A - 24C" siclog@
    // @sic R5-223430 sic@
    v_EPSBearerId := f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1); // @sic R5s220448 sic@
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_DeactivateEPS_BearerContext (eutra_Cell1, int2hex(v_EPSBearerId, 1)); // @sic R5s220448 sic@
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }



After Change:
	  function fl_TC_11_1_2_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC // @sic R5-204399 sic@
  {
    var integer v_EPSBearerId;

    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 5AA" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE);
    
    f_TAUReject_AttachWithoutN26 (eutra_Cell1);
    
    // @siclog "Parallel Behaviour" siclog@
    f_EUTRA38_MultiPDN_EstablishAdditionalPDNs (eutra_Cell1,
                                                (pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection),
                                                false,
                                                tsc_SRB2,
                                                omit,
                                                '010'B,                                                 (cr_AccessPointName(f_DomainName_Encode(pc_APN_ID_Internet)),cr_AccessPointName(f_DomainName_Encode(pc_APN_ID_IMS))),
                                                -,true);

    // @siclog "Step 24A - 24C" siclog@
    // @sic R5-223430 sic@
    v_EPSBearerId := f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1); // @sic R5s220448 sic@
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_DeactivateEPS_BearerContext (eutra_Cell1, int2hex(v_EPSBearerId, 1)); // @sic R5s220448 sic@
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  } 

  


1.2 [bookmark: _Toc122001093]Change 2
	Function name
	f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2

	Reason for change
	New parameter needs to be added to f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 for the possibility of skipping f_IP_Handling_StartPDN in case special cases like the parallel behaviour in 11.1.2.3.2-2.


	Summary of change
	Adding new parameter which optionally skips execution of f_IP_Handling_StartPDN.

	TTCN module
	Common\EUTRA38_NR5GC_InitialRegistration.ttcn

	MCC160 Comment
	



    
Before Change:
	      function f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2(EUTRA_CellId_Type p_CellId,
                                                           PDN_CONNECTIVITY_REQUEST p_PDNConnReq,
                                                           boolean p_ConfigureSRB2 := false,
                                                           template (omit) NasCount_Type p_NasCountUL := omit,
                                                           template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduInfoFromNR := omit,
                                                           boolean p_N1disabled := false) runs on EUTRA_Multi5GS_PTC return EPS_BearerIdentity
  {

…

    if (v_StartPDN) { // completely new PDN connection @sic R5s210149 sic@
      f_IP_Handling_StartPDN(IP, v_PdnIndex, cs_DrbInfo_EUTRA(p_CellId, v_NewDRBId));
    }
…




After Change:
	      function f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2(EUTRA_CellId_Type p_CellId,
                                                           PDN_CONNECTIVITY_REQUEST p_PDNConnReq,
                                                           boolean p_ConfigureSRB2 := false,
                                                           template (omit) NasCount_Type p_NasCountUL := omit,
                                                           template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduInfoFromNR := omit,
                                                           boolean p_N1disabled := false,
                                                           boolean p_ParallelBehaviour := false) runs on EUTRA_Multi5GS_PTC return EPS_BearerIdentity
  {

…

    if (p_ParallelBehaviour) {v_StartPDN := false};
    if (v_StartPDN) { // completely new PDN connection @sic R5s210149 sic@
      f_IP_Handling_StartPDN(IP, v_PdnIndex, cs_DrbInfo_EUTRA(p_CellId, v_NewDRBId));
    }
…

      


1.3 [bookmark: _Toc122001094]Change 3
	Function name
	f_EUTRA38_MultiPDN_EstablishAdditionalPDNs

	Reason for change
	New parameter needs to be added to f_EUTRA38_MultiPDN_EstablishAdditionalPDNs for the possibility to be passed down to f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 and to avoid adding wait timer for optional SIP signalling in case of IMS PDN.


	Summary of change
	Adding new parameter which is passed down to f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2 and to avoid adding wait timer for optional SIP signalling in case of IMS PDN.

	TTCN module
	Common\EUTRA38_NR5GC_InitialRegistration.ttcn

	MCC160 Comment
	



    
Before Change:
	  function f_EUTRA38_MultiPDN_EstablishAdditionalPDNs(EUTRA_CellId_Type p_CellId,
                                                      integer p_ExpectedNoOfNewPDNs,
                                                      boolean p_FailWhenK_LessN := true,
                                                      integer p_SRB := tsc_SRB2,  // @sic R5s210341 sic@
                                                      template (omit) NasCount_Type p_NasCountUL := omit,
                                                      B3_Type p_RequestType := '001'B,
                                                      template AccessPointName p_AccessPointName := cr_AccessPointName_Any,
                                                      boolean p_N1disabled := false

                                                      ) runs on EUTRA_Multi5GS_PTC return integer
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;
    var DelayForUserPlaneSignalling_Type v_WaitForUserplaneSignalling := noDelay; // Default: no IP signalling possible
    var EPS_BearerIdentity v_EpsBearerId;
    var integer v_L_NumOfReqsReceivedInParallel := 0;
    var integer v_K_NumOfPDNsProcessed := 0;
    var boolean v_EndLoop := false;
    var boolean v_RxdMsg;
    var integer v_SRB := p_SRB;
    var B3_Type v_PDNType := f_GetPdnType();
    var default v_MyDefaultVar := null;
    
    timer t_Wait := 10.0;

    // First clear store of previously received messages
    f_EUTRA38_MsgInDefault_ResetStoreOfPDNConnectivityReq();
    // Then activate store for any new msgs received
    v_MyDefaultVar := activate(a_EUTRA38_AdditionalPDNConnectivityReq(p_CellId, cr_PDNConnReqAdditionalPDN(v_PDNType, p_AccessPointName, p_RequestType))); // @sic R5-216188 sic@

    while ((v_K_NumOfPDNsProcessed < p_ExpectedNoOfNewPDNs) and not v_EndLoop) {
      v_RxdMsg := false;
      t_Wait.start;
      if (f_EUTRA38_MsgInDefault_GetNumOfPDNConnectivityReq() > v_L_NumOfReqsReceivedInParallel) {
        t_Wait.stop;
        v_L_NumOfReqsReceivedInParallel := v_L_NumOfReqsReceivedInParallel +1;
        v_ReceivedAsp := f_EUTRA38_MsgInDefault_GetOnePDNConnectivityReq(v_L_NumOfReqsReceivedInParallel);
        v_RxdMsg := true;
      } else { // no outstanding msgs, so must wait for next one
        alt {
          [] SRB.receive(car_SRB_NasPdu_IND(p_CellId, v_SRB, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNConnReqAdditionalPDN(v_PDNType, p_AccessPointName, p_RequestType))))  // @sic R5-216188 sic@
             -> value v_ReceivedAsp    /* NOTE: acc. to the prose the ESM information transfer flag is still "Not present or any allowed value"
                                        *       nevertheless if the UE sets the ESM information transfer flag to true this may cause issues
                                        *       as the generic procedure does not handle any requested ESM information transfer */
            {
              t_Wait.stop;
              v_RxdMsg := true;
            }
         [] t_Wait.timeout
            { // @sic R5-206291 sic@
              if (p_FailWhenK_LessN and (v_K_NumOfPDNsProcessed < p_ExpectedNoOfNewPDNs)) {
                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Established " & int2str(v_K_NumOfPDNsProcessed) & "PDNs instead of " & int2str(p_ExpectedNoOfNewPDNs));
              } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop
                v_EndLoop := true;
              }
            }
        }
      }
      if (v_RxdMsg) { // this will be false if either just triggered a new PDN, or about to end the loop
        // Update K now as latest msg is about to be processed
        v_K_NumOfPDNsProcessed := v_K_NumOfPDNsProcessed +1;
        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
        v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

        v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2(p_CellId, v_PDNConnReq, (v_SRB != tsc_SRB2), p_NasCountUL, -, p_N1disabled);  // @sic R5-227443 sic@
        a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(p_CellId);
        v_SRB := tsc_SRB2;  // SRB2 will have been configured by now, even if not earlier
        a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(p_CellId, v_EpsBearerId);


        select (f_EUTRA38_MultiPDN_GetPDNType(v_PDNConnReq.accessPointName)) { // @sic R5-206296 sic@
          case (IMS_DNN) {
            v_WaitForUserplaneSignalling := waitForIMS;
          }
          case (Internet_DNN) {
            v_WaitForUserplaneSignalling := dontWaitForIMS;
          }
        }
      }
    }

    // Now all PDNs on same connection have been established, wait for any signalling on user plane to finish before releasing connection
    f_EUTRA_DelayForUserPlaneSignalling (v_WaitForUserplaneSignalling);

    // all expected msgs have been processed,  so deactivate the extra default
    deactivate(v_MyDefaultVar);
    
    return v_SRB;
  }



After Change:
	  function f_EUTRA38_MultiPDN_EstablishAdditionalPDNs(EUTRA_CellId_Type p_CellId,
                                                      integer p_ExpectedNoOfNewPDNs,
                                                      boolean p_FailWhenK_LessN := true,
                                                      integer p_SRB := tsc_SRB2,  // @sic R5s210341 sic@
                                                      template (omit) NasCount_Type p_NasCountUL := omit,
                                                      B3_Type p_RequestType := '001'B,
                                                      template AccessPointName p_AccessPointName := cr_AccessPointName_Any,
                                                      boolean p_N1disabled := false,
                                                      boolean p_ParallelBehaviour := false
                                                      ) runs on EUTRA_Multi5GS_PTC return integer
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;
    var DelayForUserPlaneSignalling_Type v_WaitForUserplaneSignalling := noDelay; // Default: no IP signalling possible
    var EPS_BearerIdentity v_EpsBearerId;
    var integer v_L_NumOfReqsReceivedInParallel := 0;
    var integer v_K_NumOfPDNsProcessed := 0;
    var boolean v_EndLoop := false;
    var boolean v_RxdMsg;
    var integer v_SRB := p_SRB;
    var B3_Type v_PDNType := f_GetPdnType();
    var default v_MyDefaultVar := null;
    
    timer t_Wait := 10.0;

    // First clear store of previously received messages
    f_EUTRA38_MsgInDefault_ResetStoreOfPDNConnectivityReq();
    // Then activate store for any new msgs received
    v_MyDefaultVar := activate(a_EUTRA38_AdditionalPDNConnectivityReq(p_CellId, cr_PDNConnReqAdditionalPDN(v_PDNType, p_AccessPointName, p_RequestType))); // @sic R5-216188 sic@

    while ((v_K_NumOfPDNsProcessed < p_ExpectedNoOfNewPDNs) and not v_EndLoop) {
      v_RxdMsg := false;
      t_Wait.start;
      if (f_EUTRA38_MsgInDefault_GetNumOfPDNConnectivityReq() > v_L_NumOfReqsReceivedInParallel) {
        t_Wait.stop;
        v_L_NumOfReqsReceivedInParallel := v_L_NumOfReqsReceivedInParallel +1;
        v_ReceivedAsp := f_EUTRA38_MsgInDefault_GetOnePDNConnectivityReq(v_L_NumOfReqsReceivedInParallel);
        v_RxdMsg := true;
      } else { // no outstanding msgs, so must wait for next one
        alt {
          [] SRB.receive(car_SRB_NasPdu_IND(p_CellId, v_SRB, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNConnReqAdditionalPDN(v_PDNType, p_AccessPointName, p_RequestType))))  // @sic R5-216188 sic@
             -> value v_ReceivedAsp    /* NOTE: acc. to the prose the ESM information transfer flag is still "Not present or any allowed value"
                                        *       nevertheless if the UE sets the ESM information transfer flag to true this may cause issues
                                        *       as the generic procedure does not handle any requested ESM information transfer */
            {
              t_Wait.stop;
              v_RxdMsg := true;
            }
         [] t_Wait.timeout
            { // @sic R5-206291 sic@
              if (p_FailWhenK_LessN and (v_K_NumOfPDNsProcessed < p_ExpectedNoOfNewPDNs)) {
                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Established " & int2str(v_K_NumOfPDNsProcessed) & "PDNs instead of " & int2str(p_ExpectedNoOfNewPDNs));
              } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop
                v_EndLoop := true;
              }
            }
        }
      }
      if (v_RxdMsg) { // this will be false if either just triggered a new PDN, or about to end the loop
        // Update K now as latest msg is about to be processed
        v_K_NumOfPDNsProcessed := v_K_NumOfPDNsProcessed +1;
        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
        v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

        v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2(p_CellId, v_PDNConnReq, (v_SRB != tsc_SRB2), p_NasCountUL, -, p_N1disabled,p_ParallelBehaviour);  // @sic R5-227443 sic@
        a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(p_CellId);
        v_SRB := tsc_SRB2;  // SRB2 will have been configured by now, even if not earlier
        a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(p_CellId, v_EpsBearerId);

        if (not(p_ParallelBehaviour)) {
          select (f_EUTRA38_MultiPDN_GetPDNType(v_PDNConnReq.accessPointName)) { // @sic R5-206296 sic@
            case (IMS_DNN) {
              v_WaitForUserplaneSignalling := waitForIMS;
            }
            case (Internet_DNN) {
              v_WaitForUserplaneSignalling := dontWaitForIMS;
            }
          }
        }
      }
    }
    // Now all PDNs on same connection have been established, wait for any signalling on user plane to finish before releasing connection
    f_EUTRA_DelayForUserPlaneSignalling (v_WaitForUserplaneSignalling);

    // all expected msgs have been processed,  so deactivate the extra default
    deactivate(v_MyDefaultVar);
    
    return v_SRB;
  }
  



